clear all
set more off
cd "V:\Projects\Enum_Exp_DD\Replication_files"									//Update the path to match your local directory
global root "V:\Projects\Enum_Exp_DD\Replication_files"							//Update the path to match your local directory
capture mkdir "$root\Tables"
global tables "$root\Tables"
capture mkdir "$root\Figures"
global figures "$root\Figures"
version 14
set maxvar 20000



/*==============================================================================
				TABLE 1: SUMMARY STATISTICS AND ENUMERATOR BALANCE
==============================================================================*/

use "$root\Data\enum_rand.dta", clear



iebaltab num_language q7 q8 q14 q15 enum_pg q17 q18 q19 q20 ///
	if study_group==1, ///
	grpvar(treat_cra) ///
	savexlsx("$tables\table1_enumerator_balance.xlsx") ///
	rowvarlabels ///
	feqtest ///
	replace ///
	total ///
	order(0 1 2)	///
	addnote("Notes: This table presents summary statistics and balance checks for enumerator characteristics across treatment groups. The study recruited 60 enumerators through open advertisement and Zoom interviews in August 2023. Enumerators were block randomized based on cognitive skills and survey experience into three groups of 20 each: standard restudying (control), retesting (T1), and agentic feedback (T2). All enumerators were at least undergraduate level, could read, write, and speak Hindi, and understand basic English. The table shows mean values with standard errors in brackets. Key characteristics include number of languages spoken (average 2.2 out of 3), prior experience with child anthropometric surveys (67%), universal CAPI experience, gender composition (77% male), and average age of 34 years. Cognitive ability was assessed through multiple tests: timed basic math test (average 3.55 out of 25), digit span forward test measuring working memory (average 4.7 digits), digit span backward test (average 3.7 digits), and reading comprehension (average 2.1 out of 4). These cognitive measures are important because interviewer skills can significantly influence data quality through their ability to understand survey protocols, maintain attention, and comprehend complex questions. Over two-thirds of enumerators had experience conducting child health surveys, mainly from NFHS (National Family Health Survey) participation, and 38% held postgraduate degrees. The F-test for joint orthogonality accounts for experimental block fixed effects and shows no significant differences across treatment groups for any variable, confirming successful randomization. The highest F-statistic is 2.815 for digit span forward score (p=0.068), which remains above conventional significance levels. This balance is crucial for attributing subsequent differences in data quality outcomes to training interventions rather than pre-existing enumerator characteristics.")




* Export to LaTeX
#delimit ;
iebaltab num_language q7 q8 q14 q15 enum_pg q17 q18 q19 q20 
	if study_group==1, 
	grpvar(treat_cra) 
	savetex("$tables\table1_enumerator_balance.tex") 
	rowvarlabels 
	feqtest 
	replace 
	total 
	order(0 1 2)
	texcolwidth(3cm);
#delimit cr






/*==============================================================================
				TABLE 2: IMPACT ON DATA QUALITY 
==============================================================================*/

use "$root\Data\hh_dsy_cr_rec.dta", clear
version 14

* Standardize continuous outcomes
zscore outliers vskips invalid_skips_c inconsistency

eststo clear
local i=1

local origvars "whether_outlier outliers vskips invalid_skips_c inconsistency"

foreach var of varlist whether_outlier z_outliers z_vskips z_invalid_skips_c z_inconsistency {
	
areg `var' t1 t2 i.enum_id, a(c_strata) vce(cluster vill_id)  
test t1==t2==0
estadd scalar pval1= r(p)
test t2==t1
estadd scalar pval2= r(p)

* Get the corresponding original variable for control mean
local origvar: word `i' of `origvars'
sum `origvar' if group_code==1 & e(sample) 
estadd scalar mean=r(mean)

estadd local enum_fe "YES"
estadd local block_fe "YES"
eststo tableB1_`i'
local i=`i'+1
}

* Export to RTF (for Word)
#delimit ;
esttab tableB1_* using "$tables/table2_data_quality.rtf", replace 
    label nonumbers nogaps
    b(%9.3f) se(%9.3f) star(* 0.10 ** 0.05 *** 0.01) 
    keep(t1 t2)
    order(t1 t2)
    coeflabels(t1 "Retesting (T1)" t2 "Agentic Feedback (T2)")
    stats(enum_fe block_fe mean N pval1 pval2, 
          fmt(%~12s %~12s %9.3f %9.0g %9.3f %9.3f) 
          labels("Enumerator FE" "Experimental Block FE" 
                 "Control Mean (Absolute)" "Sample Size" 
                 "p-value: all treatments zero" "p-value: all treatments equal"))
    mtitles("Whether Data Quality Flag?" "Data Quality Flags (Z-score)" 
            "Valid Skips (Z-score)" "Invalid Missing Responses (Z-score)" 
            "Total Inconsistencies (Z-score)")
    title("TABLE 2: IMPACT ON DATA QUALITY")
    addnotes("Notes: This table presents intent-to-treat (ITT) estimates from 1,659 household surveys in rural Gujarat, India. The study randomized 60 enumerators across three training groups: (C), retesting (T1), and agentic feedback (T2). Column (1) shows treatment effects on a binary indicator equal to one if the household survey contains at least one data quality flag. Column (2) displays effects on the count of data quality flags (standardized), which identify outlier values beyond ±2 standard deviations for objective measurements including child anthropometry, household expenditures, and maternal time allocation. Column (3) presents effects on valid skips (standardized), which are variables left blank due to automated skip patterns of the questionnaire. Column (4) shows effects on invalid missing responses (standardized), which are instances where questions should have been answered but no response was recorded by the enumerator. Column (5) displays effects on inconsistencies (standardized), which refer to responses that are inconsistent with another response recorded in the survey. The agentic feedback treatment significantly reduced the likelihood of at least one data quality flag by 36.6 percentage points compared to 15 percentage points for retesting. Count of data quality flags decline by almost 0.64 SD per survey in the agentic group while showing no significant impact for retesting. Both treatments significantly reduced valid skips, though these reductions cannot be entirely attributed to training interventions because valid skips can also arise from respondent effects. Invalid missing responses are major outcome more directly associated with interviewer errors and actions compared to valid skips, providing greater confidence in attributing changes to training effects and serving as the primary data quality outcome. Agentic feedback treatment (T2) reduced invalid missing responses by 1.23 SD given the control mean of 1,217. The retesting treatment (T1) had no impact on invalid missing responses. The lack of treatment effects on inconsistencies is intuitive given the small control group (restudying) mean of 0.53. Since we are testing multiple hypotheses, we report several checks to correct for multiple hypothesis testing, including Anderson (2008)'s false discovery rate (FDR) sharpened q-values and List et al. (2019)'s family-wise error rate (FWER) p-values. All specifications include enumerator and experimental block fixed effects. Standard errors are clustered at the village level.");
#delimit cr

* Export to TEX (for LaTeX)
#delimit ;
esttab tableB1_* using "$tables/table2_data_quality.tex", replace 
    label nonumbers nogaps booktabs
    b(%9.3f) se(%9.3f) star(* 0.10 ** 0.05 *** 0.01) 
    keep(t1 t2)
    order(t1 t2)
    coeflabels(t1 "Retesting (T1)" t2 "Agentic Feedback (T2)")
    stats(enum_fe block_fe mean N pval1 pval2, 
          fmt(%~12s %~12s %9.3f %9.0g %9.3f %9.3f) 
          labels("Enumerator FE" "Experimental Block FE" 
                 "Control Mean (Absolute)" "Sample Size" 
                 "p-value: all treatments zero" "p-value: all treatments equal"))
    mtitles("Whether Data Quality Flag?" "Data Quality Flags (Z-score)" 
            "Valid Skips (Z-score)" "Invalid Missing Responses (Z-score)" 
            "Total Inconsistencies (Z-score)")
    title("TABLE 2: IMPACT ON DATA QUALITY")
    addnotes("Notes: This table presents intent-to-treat (ITT) estimates from 1,659 household surveys in rural Gujarat, India. The study randomized 60 enumerators across three training groups: (C), retesting (T1), and agentic feedback (T2). Column (1) shows treatment effects on a binary indicator equal to one if the household survey contains at least one data quality flag. Column (2) displays effects on the count of data quality flags (standardized), which identify outlier values beyond ±2 standard deviations for objective measurements including child anthropometry, household expenditures, and maternal time allocation. Column (3) presents effects on valid skips (standardized), which are variables left blank due to automated skip patterns of the questionnaire. Column (4) shows effects on invalid missing responses (standardized), which are instances where questions should have been answered but no response was recorded by the enumerator. Column (5) displays effects on inconsistencies (standardized), which refer to responses that are inconsistent with another response recorded in the survey. The agentic feedback treatment significantly reduced the likelihood of at least one data quality flag by 36.6 percentage points compared to 15 percentage points for retesting. Count of data quality flags decline by almost 0.64 SD per survey in the agentic group while showing no significant impact for retesting. Both treatments significantly reduced valid skips, though these reductions cannot be entirely attributed to training interventions because valid skips can also arise from respondent effects. Invalid missing responses are major outcome more directly associated with interviewer errors and actions compared to valid skips, providing greater confidence in attributing changes to training effects and serving as the primary data quality outcome. Agentic feedback treatment (T2) reduced invalid missing responses by 1.23 SD given the control mean of 1,217. The retesting treatment (T1) had no impact on invalid missing responses. The lack of treatment effects on inconsistencies is intuitive given the small control group (restudying) mean of 0.53. Since we are testing multiple hypotheses, we report several checks to correct for multiple hypothesis testing, including Anderson (2008)'s false discovery rate (FDR) sharpened q-values and List et al. (2019)'s family-wise error rate (FWER) p-values. All specifications include enumerator and experimental block fixed effects. Standard errors are clustered at the village level.");
#delimit cr












/*==============================================================================
				TABLE 3: IMPACT ON CHILD NUTRITION STATUS
==============================================================================*/

use "$root\Data\child_level_anthro_vill.dta", clear


eststo clear
local i=1
foreach var of varlist whether_stunted whether_wasted whether_underweight {
	
areg `var' t1 t2 if child_age_ <=60, a(c_strata) vce(cluster vill_id)   
test t1==t2==0
estadd scalar pval1= r(p)
test t2==t1
estadd scalar pval2= r(p)
sum `var' if group_code==1 & child_age_ <=60 & e(sample) 
estadd scalar mean=r(mean)
estadd local enum_fe "YES"
estadd local block_fe "YES"
eststo tableB3_`i'
local i=`i'+1
}

* Export to RTF
#delimit ;
esttab tableB3_* using "$tables/table3_child_nutrition.rtf", replace 
    label nonumbers nogaps
    b(%9.3f) se(%9.3f) star(* 0.10 ** 0.05 *** 0.01) 
    keep(t1 t2)
    order(t1 t2)
    coeflabels(t1 "Retesting (T1)" t2 "Agentic Feedback (T2)")
    stats(enum_fe block_fe mean N pval1 pval2, 
          fmt(%~12s %~12s %9.3f %9.0g %9.3f %9.3f) 
          labels("Enumerator FE" "Experimental Block FE" 
                 "Control Mean" "Sample Size" 
                 "p-value: all treatments zero" "p-value: all treatments equal"))
    mtitles("Stunted" "Wasted" "Underweight")
    title("TABLE 3: IMPACT ON CHILD NUTRITION STATUS")
    addnotes("Notes: This table presents intent-to-treat (ITT) estimates of survey training treatments on child nutrition status for children under 60 months old. The study randomized 60 enumerators across three training groups: (C), retesting (T1), and agentic feedback (T2). The sample includes 929 children under 60 months. Column (1) shows treatment effects on the probability of identifying a stunted child. Column (2) displays effects on identifying a wasted child and column (3) presents effects on identifying an underweight child. Capacity-building initiatives that enhance enumerator agency can impact their ability to approach sensitive survey sections such as child anthropometry, where enumerators must touch children which is usually inauspicious in the Indian context. The agentic feedback treatment significantly increased the probability of identifying stunted children by 11.6 pp and underweight children by 15.8 pp, both significant at the 5% level or less. The agentic feedback treatment also improved the identification of wasted children by 8.5 pp. The retesting treatment showed no significant effects on any nutrition status indicators, with point estimates statistically indistinguishable from zero. We reject the hypothesis that the effect of agentic feedback on stunting and underweight status is equivalent to the retesting treatment. Results suggest that the agentic feedback training enabled enumerators to better explain measurement procedures while respecting household beliefs. It led to more accurate identification of malnourished children. Since we are testing several hypotheses, we have reported false discovery rate (FDR) corrections and family-wise error rate (FWER) p-values. The agentic feedback effects remaining robust across corrections. All specifications include enumerator and experimental block fixed effects. Standard errors are clustered at the village level.");
#delimit cr

* Export to TEX
#delimit ;
esttab tableB3_* using "$tables/table3_child_nutrition.tex", replace 
    label nonumbers nogaps booktabs
    b(%9.3f) se(%9.3f) star(* 0.10 ** 0.05 *** 0.01) 
    keep(t1 t2)
    order(t1 t2)
    coeflabels(t1 "Retesting (T1)" t2 "Agentic Feedback (T2)")
    stats(enum_fe block_fe mean N pval1 pval2, 
          fmt(%~12s %~12s %9.3f %9.0g %9.3f %9.3f) 
          labels("Enumerator FE" "Experimental Block FE" 
                 "Control Mean" "Sample Size" 
                 "p-value: all treatments zero" "p-value: all treatments equal"))
    mtitles("Stunted" "Wasted" "Underweight")
    title("TABLE 3: IMPACT ON CHILD NUTRITION STATUS")
    addnotes("Notes: This table presents intent-to-treat (ITT) estimates of survey training treatments on child nutrition status for children under 60 months old. The study randomized 60 enumerators across three training groups: (C), retesting (T1), and agentic feedback (T2). The sample includes 929 children under 60 months. Column (1) shows treatment effects on the probability of identifying a stunted child. Column (2) displays effects on identifying a wasted child and column (3) presents effects on identifying an underweight child. All specifications include enumerator and experimental block fixed effects. Standard errors are clustered at the village level. * p < 0.10, ** p < 0.05, *** p < 0.01.");
#delimit cr






/*==============================================================================
				TABLE 4: IMPACT ON CHILD ANTHROPOMETRIC MEASURES
==============================================================================*/

use "$root\Data\child_level_anthro_vill.dta", clear


eststo clear
local i=1
foreach var of varlist haz06 waz06 whz06 {
	
areg `var' t1 t2 if child_age_ <=60 & (`var' >=-4 & `var' <=4), a(c_strata) vce(cluster enum_id)   
test t1==t2==0
estadd scalar pval1= r(p)
test t2==t1
estadd scalar pval2= r(p)
sum `var' if group_code==1 & child_age_ <=60 & e(sample) 
estadd scalar mean=r(mean)
estadd local enum_fe "YES"
estadd local block_fe "YES"
eststo tableB4_`i'
local i=`i'+1
}

* Export to RTF
#delimit ;
esttab tableB4_* using "$tables/table4_anthropometric_measures.rtf", replace 
    label nonumbers nogaps
    b(%9.3f) se(%9.3f) star(* 0.10 ** 0.05 *** 0.01) 
    keep(t1 t2)
    order(t1 t2)
    coeflabels(t1 "Retesting (T1)" t2 "Agentic Feedback (T2)")
    stats(enum_fe block_fe mean N pval1 pval2, 
          fmt(%~12s %~12s %9.3f %9.0g %9.3f %9.3f) 
          labels("Enumerator FE" "Experimental Block FE" 
                 "Control Mean" "Sample Size" 
                 "p-value: all treatments zero" "p-value: all treatments equal"))
    mtitles("HAZ" "WAZ" "WHZ")
    title("TABLE 4: IMPACT ON CHILD ANTHROPOMETRIC MEASURES")
    addnotes("Notes: This table presents intent-to-treat (ITT) estimates of survey training treatments on child anthropometric scores for children under 60 months old. The study randomized 60 enumerators across three training groups: standard restudying (C), retesting (T1), and agentic feedback (T2). Children less than 60 months are used in the analysis. Sample sizes vary by outcome due to measurement availability (773 for HAZ, 822 for WAZ, 813 for WHZ). Column (1) shows treatment effects on height-for-age z-scores (HAZ). Column (2) displays effects on weight-for-age z-scores (WAZ). Column (3) presents effects on weight-for-height z-scores (WHZ). The impact of training interventions on nutrition z-scores is aligned with the nutrition status estimates in Table 3. The retesting treatment does not significantly affect any nutrition z-scores except for weight-for-height, which is lower by 0.305 standard deviations. The agentic feedback treatment significantly reduced HAZ by 0.466 standard deviations and WAZ by 0.405 standard deviations compared to the control group. We reject the null of treatment equivalence for both HAZ and WAZ. The agentic feedback treatment had a statistically significant effect on nutritional measures, because enumerators from T2 are more likely to identify stunted and malnourished children compared to T1. When measuring child growth parameters, enumerators are required to touch children which is usually considered inauspicious in the Indian context. Findings suggest that agentic training enabled enumerators to better explain procedures while respecting household beliefs and sentiments. The table also reports results from false discovery rate (FDR) corrections and family-wise error rate (FWER) p-values. All specifications include enumerator and experimental block fixed effects. Standard errors are clustered at the village level.");
#delimit cr

* Export to TEX
#delimit ;
esttab tableB4_* using "$tables/table4_anthropometric_measures.tex", replace 
    label nonumbers nogaps booktabs
    b(%9.3f) se(%9.3f) star(* 0.10 ** 0.05 *** 0.01) 
    keep(t1 t2)
    order(t1 t2)
    coeflabels(t1 "Retesting (T1)" t2 "Agentic Feedback (T2)")
    stats(enum_fe block_fe mean N pval1 pval2, 
          fmt(%~12s %~12s %9.3f %9.0g %9.3f %9.3f) 
          labels("Enumerator FE" "Experimental Block FE" 
                 "Control Mean" "Sample Size" 
                 "p-value: all treatments zero" "p-value: all treatments equal"))
    mtitles("HAZ" "WAZ" "WHZ")
    title("TABLE 4: IMPACT ON CHILD ANTHROPOMETRIC MEASURES");
#delimit cr







/*==============================================================================
				TABLE 5: IMPACT ON CALORIE, MACRONUTRIENTS, AND FOOD DIVERSITY
						(WITHOUT FOOD DIVERSITY)
==============================================================================*/

use "$root\Data\child_calorie.dta", clear

eststo clear
local i=1
foreach var of varlist ch_calories_kcal_win_top20 ch_protein_win_top20 ch_fat_win_top20 ch_carbohydrates_win_top20 {
	
areg `var' t1 t2 child_age hh_income i.education_father i.education_mother, a(c_strata) vce(cluster vill_id)  
test t1==t2==0
estadd scalar pval1= r(p)
test t2==t1
estadd scalar pval2= r(p)
estadd local enum_fe "YES"
estadd local block_fe "YES"
estadd local child_control "YES"
estadd local parental_control "YES"
estadd local hh_control "YES"
eststo tableB5_`i'
local i=`i'+1
}

* Export to RTF
#delimit ;
esttab tableB5_* using "$tables/table5_consumption_nutrients.rtf", replace 
    label nonumbers nogaps
    b(%9.3f) se(%9.3f) star(* 0.10 ** 0.05 *** 0.01) 
    keep(t1 t2)
    order(t1 t2)
    coeflabels(t1 "Retesting (T1)" t2 "Agentic Feedback (T2)")
    stats(enum_fe block_fe child_control parental_control hh_control N pval1 pval2, 
          fmt(%~12s %~12s %~12s %~12s %~12s %9.0g %9.3f %9.3f) 
          labels("Enumerator FE" "Experimental Block FE" "Child Control" "Parental Control" "HH Control"
                 "Sample Size" 
                 "p-value: all treatments zero" "p-value: all treatments equal"))
    mtitles("Daily Calorie" "Protein" "Fat" "Carbohydrate")
    title("TABLE 5: IMPACT ON CALORIE, MACRONUTRIENTS, AND FOOD DIVERSITY")
    addnotes("Notes: This table presents intent-to-treat (ITT) estimates of survey training treatments on child food consumption measures based on 24-hour dietary recall. The study randomized 60 enumerators across three training groups: standard restudying (C), retesting (T1), and agentic feedback (T2). Enumerators tend to strategically club multiple food items together while canvassing the consumption section, resulting in under-reporting of consumed food items. Therefore, values of protein, calories, fat, carbohydrates, and food diversity are expected to be positively correlated with the extent of completeness of the survey data. Consumption data for all children has been used in the analysis. Sample sizes vary by outcome (2,667 for food diversity, calories, and protein; 2,132 for fat and carbohydrates). Column (1) shows treatment effects on food diversity score, calculated by summing unique food items consumed across four meals (breakfast, lunch, snacks, dinner) from 107 possible food items. Column (2) displays effects on daily calorie intake (kcal). Column (3) presents effects on protein consumption (grams). Column (4) shows effects on fat intake (grams) and column (5) shows effect on carbohydrate consumption (grams). The consumption section tracked child consumption across four daily meals with a food list including 107 items containing processed foods, nutrition supplements, and local cuisines. Training treatments were expected to influence child consumption through reduction in valid skips (more food items recorded) and enhanced interviewer probing capacities. The food diversity score is 1.363 higher in the retesting group and 1.708 higher in the agentic feedback group compared to control. Calorie intake values are 117.84 kcal higher for retesting and 77.40 kcal higher for agentic feedback. Retesting treatment improved the reported protein consumption by 19.09 grams and agentic treatment by 10.54 grams. Fat and carbohydrate intake are also higher in the agentic feedback group, though fat intake in the retesting group is not statistically different from the control group. The positive impact reflects increased food diversity scores from reduction in valid skips leading to reporting of more food items. Since we are testing multiple hypotheses, we reported false discovery rate (FDR) corrections and family-wise error rate (FWER) p-values. All specifications include enumerator and experimental block fixed effects plus child, parental, and household controls. Standard errors are clustered at the village level.");
#delimit cr

* Export to TEX
#delimit ;
esttab tableB5_* using "$tables/table5_consumption_nutrients.tex", replace 
    label nonumbers nogaps booktabs
    b(%9.3f) se(%9.3f) star(* 0.10 ** 0.05 *** 0.01) 
    keep(t1 t2)
    order(t1 t2)
    coeflabels(t1 "Retesting (T1)" t2 "Agentic Feedback (T2)")
    stats(enum_fe block_fe child_control parental_control hh_control N pval1 pval2, 
          fmt(%~12s %~12s %~12s %~12s %~12s %9.0g %9.3f %9.3f) 
          labels("Enumerator FE" "Experimental Block FE" "Child Control" "Parental Control" "HH Control"
                 "Sample Size" 
                 "p-value: all treatments zero" "p-value: all treatments equal"))
    mtitles("Daily Calorie" "Protein" "Fat" "Carbohydrate")
    title("TABLE 5: IMPACT ON CALORIE, MACRONUTRIENTS, AND FOOD DIVERSITY");
#delimit cr







/*==============================================================================
				TABLE 6: PSYCHOLOGICAL WELLBEING
==============================================================================*/

use "$root\Data\hh_dsy_cr_rec.dta", clear
version 14


* Standardize Cohen's stress scale items
zscore sec6_q47 sec6_q48 sec6_q49 sec6_q50 sec6_q51 sec6_q52 sec6_q53 sec6_q54 sec6_q55 sec6_q56

* Stress Index 1 - negative emotions (items with negative connotations)
* Items: upset by unexpected, unable to control, nervous/stressed, could not cope, angered by uncontrollable, difficulties piling up

gen stress_index1 = (z_sec6_q47 + z_sec6_q48 + z_sec6_q49 + z_sec6_q52 + z_sec6_q55 + z_sec6_q56)/6   
label var stress_index1 "Stress Index 1 (negative emotions)"

* Stress Index 2 - positive emotions (items with positive connotations)
* Items: confident handling problems, things going her way, controlling irritations, feelings on top of things

gen stress_index2 = (z_sec6_q50 + z_sec6_q51 + z_sec6_q53 + z_sec6_q54)/4
label var stress_index2 "Stress Index 2 (positive emotions)"

eststo clear
local i=1
foreach var of varlist stress_index1 stress_index2 var_wellbeing {
	
areg `var' t1 t2 i.enum_id, a(c_strata) vce(cluster vill_id) 
test t1==t2==0
estadd scalar pval1= r(p)
test t2==t1
estadd scalar pval2= r(p)
sum `var' if group_code==1 & e(sample)
estadd scalar mean=r(mean)
estadd local enum_fe "YES"
estadd local block_fe "YES"
eststo tableB6_`i'
local i=`i'+1
}

* Export to RTF
#delimit ;
esttab tableB6_* using "$tables/table6_psychological_wellbeing.rtf", replace 
    label nonumbers nogaps
    b(%9.3f) se(%9.3f) star(* 0.10 ** 0.05 *** 0.01) 
    keep(t1 t2)
    order(t1 t2)
    coeflabels(t1 "Retesting (T1)" t2 "Agentic Feedback (T2)")
    stats(enum_fe block_fe mean N pval1 pval2, 
          fmt(%~12s %~12s %9.3f %9.0g %9.3f %9.3f) 
          labels("Enumerator FE" "Experimental Block FE" 
                 "Control Mean" "Sample Size" 
                 "p-value: all treatments zero" "p-value: all treatments equal"))
    mtitles("Stress Index 1" "Stress Index 2" "Variance in Psychological Wellbeing")
    title("TABLE 6: PSYCHOLOGICAL WELLBEING")
    addnotes("Notes: This table presents intent-to-treat (ITT) estimates of survey training treatments on maternal psychological wellbeing measures. The study randomized 60 enumerators across three training groups: standard restudying (C), retesting (T1), and agentic feedback (T2). Column (1) shows treatment effects on Stress Index 1, which is computed by taking the average z-scores of six Cohen's stress scale items associated with negative emotions (upset by unexpected events, unable to control important things, nervous and stressed, could not cope, angered by uncontrollable events, difficulties piling up). It measured on ordinal scale from never to very often. Column (2) displays effects on Stress Index 2, which is constructed by taking the average z-scores of four Cohen's stress scale items with positive emotional connotations (confident handling problems, things going her way, controlling irritations, feelings on top of things), measured on the same ordinal scale. Column (3) presents effects on variance in psychological wellbeing responses across 21 items including the 10 Cohen's stress scale items plus 11 additional wellbeing indicators. Measurement of psychological wellbeing is complex due to scale items that may appear similar, requiring enumerators to understand distinctions and explain items clearly to respondents. Negative estimates for Stress Index 1 suggest improved psychological wellbeing, while positive estimates for Stress Index 2 indicate improved wellbeing. Both treatments show large magnitude effects statistically different from control. Retesting treatment reduced Stress Index 1 by 1.307 and increased Stress Index 2 by 2.769. Agentic feedback treatment reduced Stress Index 1 by 2.386 and increased Stress Index 2 by 1.300. One major source of measurement error is inability to communicate nuanced distinctions among scale items, which reduces actual variation in responses. The variance in wellbeing responses increased by 0.666 for retesting but decreased by 0.429 for agentic feedback, both statistically different from control and from each other. Since we are testing multiple hypotheses, we report false discovery rate (FDR) corrections and family-wise error rate (FWER) p-values. All specifications include enumerator and experimental block fixed effects. Standard errors are clustered at the village level.");
#delimit cr

* Export to TEX
#delimit ;
esttab tableB6_* using "$tables/table6_psychological_wellbeing.tex", replace 
    label nonumbers nogaps booktabs
    b(%9.3f) se(%9.3f) star(* 0.10 ** 0.05 *** 0.01) 
    keep(t1 t2)
    order(t1 t2)
    coeflabels(t1 "Retesting (T1)" t2 "Agentic Feedback (T2)")
    stats(enum_fe block_fe mean N pval1 pval2, 
          fmt(%~12s %~12s %9.3f %9.0g %9.3f %9.3f) 
          labels("Enumerator FE" "Experimental Block FE" 
                 "Control Mean" "Sample Size" 
                 "p-value: all treatments zero" "p-value: all treatments equal"))
    mtitles("Stress Index 1" "Stress Index 2" "Variance in Psychological Wellbeing")
    title("TABLE 6: PSYCHOLOGICAL WELLBEING");
#delimit cr


 
	   
	   
	   
	   
	   



